Linq to SQL

Linq es el nuevo ORM – Object Relational Model que trata de saltar la brecha existente entre dos mundos otrora incompatibles: Los lenguajes de programación y los lenguajes de acceso a bases de datos. 
Antes de Linq, los lenguajes de programación no tenían noción de los datos, y por ello nos encontrábamos escribiendo sentencias SQL en strings, sin oportunidad que los lenguajes nos pudieran ayudar con intellisense (que automáticamente liste tablas, campos, etc.) y con revisión de sintaxis, etc.
Creando el Modelo
Asumimos que tenemos ya un proyecto. En nuestro ejemplo, es un proyecto Web (conocidos en el mundo .NET como ASP.NET). 
En el Solution Explorer dar click derecho sobre el nombre del proyecto y pida la opción Add New Item…


Escoger el ítem: LINQ to SQL Classes
y en el nombre (Name) póngale un nombre descriptivo, como Northwind.dbml
Presione en Add.

Aparece un mensaje indicando que insertará el archivo dentro de la carpeta App_Code. Conteste que Sí. Aparece el Object Relational Designer.

Note que a la derecha del Object Relational Designer hay una zona vertical para insertar métodos. Si esta zona está escondida y desea verla, puede encenderla dando click derecho sobre el Object Relational Designer y pidiendo la opción Show Methods Pane.

Acá puede crear entidades y relaciones manualmente, o los puede crear a partir de tablas de la base de datos, yendo al Server Explorer y arrastrando tablas al área de diseño.

Note que el diseñador trata de convertir los nombres de tabla en singular: Products se convierte en Product. Si desea cambiar el nombre, de doble click sobre el nombre.

Es posible también insertar objetos como Vistas.
En este ejemplo, hemos adicionado la vista Invoices:

Si quiere agregar Procedimientos Almacenados (Stored Procedures), arrástrelo hacia el área de métodos. En este ejemplo, hemos arrastrado el procedimiento almacenado llamado Ten Most Expensive Products (Los 10 Productos Más Caros) hacia el área de métodos.

Recuerde que el área debe estar visible con ShowMethods Pane.

Note que hay una relación entre Product y Category. Automáticamente en el DataContext genera una propiedad Category dentro del objeto Product, de forma que podemos accesar a Product.Category.CategoryName. Esta relación puede ser modificada al seleccionarla. Note que la Cardenality es OneToMany (Uno a muchos).

Uno de los ajustes que podemos hacer es que sólo carguen imágenes cuando son necesarios, para ahorrar ancho de banda. Por ejemplo, el campo Picture de la tabla Category:

Si quisiera usar Stored Procedures para insertar, modificar o eliminar, acá mostramos dónde deben especificarse:

Para especificar un Stored Procedure, escoja la opción Customize

Modificaremos nuestro DataContext para la siguiente demostración:

Y en la página web Default.aspx dejaremos caer un GridView:

Le puse un nombre más bonito al GridView: gvNorthwind
El código queda así:


Partial Class _Default
    Inherits System.Web.UI.Page

    Protected Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles Me.Load

        Dim db As New NorthwindDataContext

        ' product tiene la interface iEnumerable,
        ' que permite DataBinding
        Dim product = From p In db.Products _
                      Select p

        gvNorthwind.DataSource = product
        gvNorthwind.DataBind()


    End Sub

End Class


Lo corremos (posiblemente aparezca un mensaje pidiendo cambiar el Web Config para poder debug. Contestemos que sí):

Modificamos nuestra expresión LINQ para que sólo muestre los productos cuya categoría tenga un nombre específico. Note que este tipo de query implica un JOIN entre dos tablas.
Dim product = From p In db.Products _
              Where p.Category.CategoryName = "Beverages" _
              Select p
Pondremos un breakpoint y vamos a correr de nuevo la aplicación:

Una vez interrumpidos por el breakpoint, podemos ver los datos devueltos:

Supuestamente se tendría que ver también el Query, y aparece una lupa que hay que darle click. En el video, aparece la siguiente ventana:

El resultado de correr la aplicación es la siguiente:

Modificaremos nuestra expresión LINQ para mostrar otras capacidades:
Protected Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles Me.Load

        Dim db As New NorthwindDataContext

        ' product tiene la interface iEnumerable,
        ' que permite DataBinding
        Dim product = From p In db.Products _
                      Select p.ProductID, _
                      p.ProductName, _
                      numOrders = p.Order_Details.Count, _
                      revenue = Aggregate detail In p.Order_Details _
                                Into Sum(detail.UnitPrice * detail.Quantity)


        gvNorthwind.DataSource = product
        gvNorthwind.DataBind()


    End Sub
Note que estamos haciendo un .Count de los Order_Details de p.
Y luego, hacemos un aggregate usando p.Order_Details sumando la multiplicación del precio unitario con la cantidad, para tener el valor total de un producto.

Nota: No sé por qué el orden de las columnas sale de esa manera. Tal vez haya que especificar directamente en el GridView las columnas que queremos ver.
Vamos a Paginar en el Server, pues hay muchos resultados. 
En el siguiente ejemplo vamos a la fila 50 y tomamos 10 registros.
        gvNorthwind.DataSource = product.Skip(50).Take(10)
        gvNorthwind.DataBind()


Vamos a mover el código de acceso hacia una rutina aparte, para que lo podamos llamar desde varias partes. Parametrizamos la fila inicial con un parámetro llamado startRow
Protected Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles Me.Load

        Dim row As Integer = Convert.ToInt32(Request.QueryString("startRow"))
        bindProducts(row)

End Sub


Protected Sub bindProducts(ByVal startRow As Integer)

        Dim db As New NorthwindDataContext

        ' product tiene la interface iEnumerable,
        ' que permite DataBinding
        Dim product = From p In db.Products _
                      Select p.ProductID, _
                      p.ProductName, _
                      numOrders = p.Order_Details.Count, _
                      revenue = Aggregate detail In p.Order_Details _
                                Into Sum(detail.UnitPrice * detail.Quantity)


        gvNorthwind.DataSource = product.Skip(startRow).Take(10)
        gvNorthwind.DataBind()

End Sub
El resultado, si ponemos un Query String así es:
http://localhost/DevCare/WebSiteLinqTest20080204/Default.aspx?startRow=50

Parecería aca que primero se hace el Select trayendo todos los datos, y luego se pagina.
Sin embargo, LINQ utiliza un modelo de Deferred Execution (Ejecución Diferida), que le permite ser más eficiente.


Modificaremos el Mapping File (para el Data Context) ligeramente para la siguiente demostración:

Agregamos 3 botones al formulario ASP.NET. Les puse nombres bonitos como: btnLame, etc.

Queremos buscar productos caros que no se están vendiendo bien (Productos Lame).
El SQL para dicha operación es el siguiente:
Select p.ProductID, p.ProductName, p.UnitPrice, p.ReorderLevel, sum(od.Quantity) as "UnitsSold"
from Products as p, "Order Details" as od
where od.ProductID = p.ProductID
and p.UnitPrice > 15
group by p.ProductID, p.ProductName, p.UnitPrice, p.ReorderLevel
having sum(od.Quantity) < 200
order by UnitSold
En nuestro código ASP.NET para el botón btnLame escribimos el siguiente código:
    Protected Sub btnLame_Click(ByVal sender As Object, _
    ByVal e As System.EventArgs) Handles btnLame.Click

        Dim db As New NorthwindDataContext

        Dim r = From p In db.Products _
                Where p.UnitPrice > 15 And _
                p.Order_Details.Sum(Function(x) x.Quantity) < 200 _
                Select p

        gvNorthwind.DataSource = r
        gvNorthwind.DataBind()

    End Sub
Al correrlo y oprimir el botón Lame tenemos los productos caros que no tienen mucho movimiento:







Vamos actualizar los ReorderLevel a cero, pues NO queremos pedir más de estos productos.
    Protected Sub btnLame_Click( _
    ByVal sender As Object, _
    ByVal e As System.EventArgs) Handles btnLame.Click

        Dim db As New NorthwindDataContext

        Dim r = From p In db.Products _
                Where p.UnitPrice > 15 And _
                p.Order_Details.Sum(Function(x) x.Quantity) < 200 _
                Select p

        For Each p As Product In r
            p.ReorderLevel = 0
        Next

        db.SubmitChanges()

        gvNorthwind.DataSource = r
        gvNorthwind.DataBind()

    End Sub

Y vemos que todos los ReorderLevel están en cero.
Un punto interesante es que LINQ sólo envió dos Update, pues el producto con ProductID=9 ya tenía cero desde antes y no era necesario actualizarlo.
Nota: Usa optimistic concurrency, por lo que el WHERE lleva todos los campos.
Transacciones

Antes de usar las transacciones, hay que agregar una referencia a System.Transactions.
Botón derecho sobre el proyecto, y pedir la opción de Property Pages.
Colóquese en References y oprima el botón Add…
Agregue System.Transactions




Y modificamos nuestro código:
Protected Sub btnLame_Click(ByVal sender As Object, ByVal e As System.EventArgs) Handles btnLame.Click

        Using ts As New Transactions.TransactionScope

            Dim db As New NorthwindDataContext

            Dim r = From p In db.Products _
                    Where p.UnitPrice > 15 And _
                    p.Order_Details.Sum(Function(x) x.Quantity) < 200 _
                    Select p

            db.Products.DeleteAllOnSubmit(r)
            db.SubmitChanges()

            gvNorthwind.DataSource = r
            gvNorthwind.DataBind()

            Transactions.Transaction.Current.Rollback()

        End Using


    End Sub
Ponemos un breakpoint en el Transactions.Transaction.Current.Rollback() y cuando lo corremos y se queda trabado en el breakpoint, si vamos al SQL, no podemos ver los registros, pues están enllavados (locked).
En la base de datos SQL podemos poner:
set transaction isolation level read uncommitted

que nos permite leer “dirty records” y al correr un query veremos que los registros parece que ya han sido borrados.


Crear Registros

Crearemos un nuevo Producto.
Protected Sub btnNewProduct_Click( _
        ByVal sender As Object, _
        ByVal e As System.EventArgs) Handles btnNewProduct.Click

        Dim db As New NorthwindDataContext

        Dim p As Product = New Product
        p.UnitPrice = -4.3

        db.Products.InsertOnSubmit(p)
        db.SubmitChanges()



    End Sub
Esto nos da un error, pues el ProductName no puede ser nulo. Lo interesante es que el error es detectado en el lado del cliente, y no por el servidor.

Este código sí funciona, pues pasa todos los Constraints:
Protected Sub btnNewProduct_Click( _
        ByVal sender As Object, _
        ByVal e As System.EventArgs) Handles btnNewProduct.Click

        Dim db As New NorthwindDataContext

        Dim p As Product = New Product
        p.ProductName = "Cosito"
        p.UnitPrice = 4.3

        db.Products.InsertOnSubmit(p)
        db.SubmitChanges()

    End Sub


Este código hace una inserción más complicada:
   Protected Sub btnNewOrder_Click( _
        ByVal sender As Object, _
        ByVal e As System.EventArgs) Handles btnNewOrder.Click

        Dim db As New NorthwindDataContext

        Dim chai As Product = _
            db.Products.Single(Function(p) p.ProductName = "Chai")
        Dim tofu As Product = _
            db.Products.Single(Function(p) p.ProductName = "Tofu")

        ' Nota: No tenemos que preocuparnos por AutoIncrement
        Dim o As New Order
        o.OrderDate = DateTime.Now
        o.RequiredDate = DateTime.Now.AddDays(2)
        o.Freight = 4.3

        Dim li1 As New Order_Detail With {.Product = chai, .Quantity = 23}
        Dim li2 As New Order_Detail With {.Product = tofu, .Quantity = 33}

        o.Order_Details.Add(li1)
        o.Order_Details.Add(li2)

        Dim c As Customer = _
           db.Customers.Single(Function(x) x.CompanyName.StartsWith("Alfred"))
        c.Orders.Add(o)

        db.SubmitChanges()


    End Sub



Reglas del Negocio

En el Solution Explorer, damos click derecho sobre el proyecto y pedimos Add New Item…
Pedimos un Class que le pondremos Order.cls

La clase resultante le ponemos el calificativo partial, pues extenderá a una clase existente.
Imports Microsoft.VisualBasic

Partial Public Class Order

    Private Sub OnFreightChanging(ByVal value As Decimal?)
        If value <= 0 Then
            Throw New ApplicationException("No free shipping")
        End If
    End Sub

    Private Sub OnValidate(ByVal action As System.Data.Linq.ChangeAction)
        If RequiredDate < OrderDate Then
            Throw New ApplicationException("Time travel not possible")
        End If
    End Sub

End Class
Modificamos nuestro código que agrega la orden, para que rompa las reglas del negocio:
:
        Dim o As New Order
        o.OrderDate = DateTime.Now
        o.RequiredDate = DateTime.Now.AddDays(-2)
        o.Freight = -4.3
:

Si corremos la aplicación y pedimos agregar la orden:

Si corregimos el Shipping, nos da el error de la fecha:
Note que el lugar en donde da error varía con respecto al anterior.



Las 4 funciones: Select, Insert, Update, Delete

Agregamos cuatro botones para mostrar la funcionalidad:

El código para el botón Select será este:
    Protected Sub btnSelect_Click( _
        ByVal sender As Object, _
        ByVal e As System.EventArgs) Handles btnSelect.Click

        showProducts()

    End Sub

    Protected Sub showProducts()
        Dim db As New NorthwindDataContext

        Dim products = From p In db.Products _
                       Where p.Category.CategoryName = "Beverages" _
                       Order By p.ProductID Descending _
                       Select New _
                       With {p.ProductID, _
                             p.ProductName, _
                             p.Category.CategoryName, _
                             p.UnitPrice}

        gvNorthwind.DataSource = products
        gvNorthwind.DataBind()

    End Sub
Y el resultado es éste:

Ahora insertaremos un registro:
Protected Sub btnInsert_Click( _
        ByVal sender As Object, _
        ByVal e As System.EventArgs) Handles btnInsert.Click

        Dim db As New NorthwindDataContext
        Dim p As New Product _
            With { _
                .ProductName = "Scott's Apple Juice", _
                .UnitPrice = 2.4, _
                .CategoryID = 1}

        db.Products.InsertOnSubmit(p)
        db.SubmitChanges()

        showProducts()
    End Sub
Al correrlo, y oprimir Insert tenemos:
(en la página resultante, vemos el nuevo registro al inicio: “Scott’s Apple Juice”)


Actualizando un registro:
    Protected Sub btnUpdate_Click( _
        ByVal sender As Object, _
        ByVal e As System.EventArgs) Handles btnUpdate.Click

        Dim db As New NorthwindDataContext

        Dim p1 = (From p In db.Products _
                Where p.ProductName.StartsWith("Scott's") _
                Select p).Single
        p1.UnitPrice = p1.UnitPrice * 1.5

        db.SubmitChanges()

        showProducts()

    End Sub

El resultado es que Scott’s Apple Juice ya no cuesta 2.40 sino 3.60-


Ahora vamos a eliminar:
    Protected Sub btnDelete_Click( _
        ByVal sender As Object, _
        ByVal e As System.EventArgs) Handles btnDelete.Click

        Dim db As New NorthwindDataContext

        Dim p1 = db.Products.First( _
            Function(x) x.ProductName.StartsWith("Scott's"))
        db.Products.DeleteOnSubmit(p1)
        db.SubmitChanges()

        showProducts()

    End Sub




Al correrlo y oprimir Delete tenemos:

Ya no está el Scott’s Apple Juice.



LinqDataSource

Esta vez tendremos el acceso de datos en una clase aparte. Cree un proyecto tipo ClassLibrary que le puse: clsLibData20080211. Con File – New Project…

No usaremos el archivo que nos da por default: Class1.vb (podríamos incluso borrarlo).
Mas bien, vamos a agregar una clase Linq to SQL Data Context.
En el Solution Explorer, damos click derecho sobre el nombre del proyecto clsLibData20080211 y pedimos Add – New Item…

(Le puse: dcNorthwind.dbml)

Desde el Server Explorer arrastré las tablas Orders, Order Details, Products, Categories, Customers. Note que puede acercarse y alejarse del diagrama de tablas utilizando Ctrl + Scroll Wheel del ratón.
Damos un build al proyecto y nos olvidaremos por el momento de él.


Proyecto Web

Ahora crearemos otro proyecto, tipo ASP.NET (Web) para interfaz del usuario.
Vamos a File – Add – New Web Site…

Referenciamos el proyecto anterior clsLibData20080211, que es la que accesa los datos. En el Solution Explorer damos click derecho sobre nuestro proyecto Web: 

En la página que nos da al inicio: Default.aspx agregamos un GridView. Yo lo nombré como gvProducts. 



Damos click en la flechita en la parte superior derecha del GridView para tener el menú contextual del GridView. Pedimos la opción: Choose Data Source - <New Data Source…>

Yo le puse de nombre: ldsNorthwind y luego, oprimimos OK.


Si no aparece el dcNorthwindDataContext, probablemente nos falte la referencia a la clase clsLibData20080211, o nos falta hacer un Build de nuestro proyecto Web.


Escogemos la tabla Products para este ejemplo.

Para nuestro ejemplo, también oprimimos el botón Advanced… en donde le diremos que nuestro Data Source permitirá actualizaciones a los datos:



Al regresar al Asistente, damos Finish y ya tenemos el Linq Data Source adicionado a la página Web:

En el menú contextual del GridView podemos ahora habilitar paginación, ordenamiento y hasta edición. Tenemos las opciones de edición debido a que en el LinqDataSource especificamos en Advanced que permitiera agregar, editar o eliminar.

Vamos a eliminar algunas de las columnas que se muestran al usuario.
Para ello, debemos marcar la columna y en el menú contextual del GridView (no de la Columna), pedimos Remove Column.

Para nuestro ejercicio, eliminamos las columnas UnitPrice y UnitsInStock.
Le dimos un AutoFormat para que se vea mejor:
[image: ]
Si lo corremos, tenemos la página Web que muestra los datos.
[image: ]
Vemos que podemos editar (oprimiendo el link Editar), paginar, ordenar:
[image: ]
Sin embargo, vemos que SupplierID y CategoryID son Foreign Key que no son nada amigables para el usuario.
Eliminaremos estas dos columnas, pero esta vez desde el código fuente HTML de la página.
Los marcamos y borramos:
[image: ]



En vez de ello, pondremos 2 template columns:

<asp:TemplateField HeaderText="Category" SortExpression="Category.CategoryName">
    <ItemTemplate>
        <%#Eval("Category.CategoryName")%>
    </ItemTemplate>
</asp:TemplateField>

<asp:TemplateField HeaderText="Supplier" SortExpression="Supplier.CompanyName">
    <ItemTemplate>
        <%#Eval("Supplier.CompanyName")%>
    </ItemTemplate>
</asp:TemplateField>

El resultado es que ahora vemos los nombres de la Categoría (Beverages, Condiments, Produce…) y del Proveedor (Exotic Liquids, New Orleans Cajun Delights, Grandma Kelly’s Homestead…). Sin embargo, no podemos editarlos:

Los Joins son manejados automáticamente por Linq.



Ahora modificaremos los Template Columns para permitir la modificación.
Seleccionamos la columna que queremos modificar (comenzaremos con Category) NOTA: Creo que es indiferente qué columna marquemos, pues más tarde nos pregunta por la columna. Damos botón derecho, Edit Template y escogemos la columna: Column[3] - Category
[image: ]
Nos interesa el EditItemTemplate, que muestra lo que verá el usuario cuando esté editando.
Arrastramos un DropDownList (un combo box). Yo le puse el nombre: ddlCategory.

Vamos a necesitar un nuevo DataSource para el Drop Down List. Pedimos el menú contextual del Drop Down List, Choose Data Source…. En la caja de diálogo, donde pide Select a Data Source: pedimos <New Data Source…>.

Yo le puse de ID: ldsCategory.
Pedimos el Context Object que reside en nuestro Class Library clsLibData20080211:





Escogemos la tabla Categories. 
Sólo vamos a querer los campos: CategoryID y CategoryName:

Damos Finish y regresamos al Asistente de Choose a Data Source…
Escogemos cuál campo se va a mostrar: CategoryName, y cuál campo tiene los valores a devolver: CategoryID,  por el Drop Down List ddlCategory:



Regresamos a la edición del Template. Tenemos ahora que especificar los Data Bindings, de forma que lo mostrado en el Drop Down List ddlCategory sea aplicado a la tabla Products del GridView. En el menú contextual de ddlCategory pedimos la opción: Edit DataBindings.


Para salir del modo de edición del Template, en el menú contextual del Template hay una opción End Template Editing

Haremos el mismo proceso con el Supplier.



Al correrlo y pedir Editar tenemos lo siguiente:

Hoy ya podemos cambiar la categoría y el proveedor (supplier) escogiendo de drop down lists.
Filtrando por Categorías

Ahora vamos a filtrar el grid a partir de un drop down list de categorías. El grid sólo mostrará los productos de la categoría escogida en el drop down list.
Agregaremos un drop down list inmediatamente debajo de donde comienza el formulario. Lo haremos dentro del código HTML:
   <form id="form1" runat="server">
    <div>
       Escoja una Categoría:
        <asp:DropDownList ID="ddlCategoryFilter" runat="server">
        </asp:DropDownList>       
    </div>

Al Drop Down List le puse el nombre: ddlCategoryFilter


Usaremos el mismo LinqDataSource de Category ldsCategory que agregamos para el drop down list cuando estamos editando el grid. Sin embargo, el scope del data source es sólo para el template. Lo sacaremos de allí y lo pondremos fuera, donde puede ser usado por cualquier control del formulario.
NOTA: Yo no estoy muy seguro que sea conveniente re-utilizar Data Sources. A mí me pasaba con Windows applications que si dos controles usaban el mismo Data Source, ambos controles estaban conectados entre sí. Tal vez en el Web este problema no se da.

Borramos el asp:LinqDataSource del Template, y lo colocamos en otra parte, casi al final del formulario:



Ahora regresamos a nuestr Drop Down List ddlCategoryFilter. En modo de diseño, pedimos su menú contextual – Choose a Data Source…

Nota: Es posible que tenga que oprimir Refresh Schema para que aparezcan los campos para desplegar y para escoger el valor.

Para que inmediatamente al cambiar de categoría nos muestre la página actualizada, pediremos: Enable AutoPostBack (Otra opción sería poner un botón que lleve a cabo la acción).
También tenemos que indicarle al Grid que sólo muestre los productos de una categoría específica.
Marcamos el Grid, vamos a su menú contextual y pedimos: Configure Data Source… Damos Next para pasar a la pantalla siguiente, donde nos muestra la tabla y los campos a mostrar. Oprimimos el botón Where…

Queremos que la columa CategoryID sea igual (==) al Control ddlCategoryFilter.
No se le olvide oprimir en Add para adicionar la expresión.




Hoy ya muestra los productos según la categoría elegida.





Stored Procedures (Procedimientos Almacenados)

Para este ejemplo añadiremos un LINQ to SQL class, que yo llamé dcCustOrderList.dbml.

Abrimos el area de diseño de dcCustOrderList.dbml. Note que a la derecha hay un área para métodos. Si no se muestra, puede encenderla con botón derecho sobre el area de diseño, Show Methods Pane. Arrastraremos también la tabla Products para usarlo en el ejercicio.

Vamos a trabajar con el Stored Procedures llamado CustOrderHist que tiene el siguiente código:

ALTER PROCEDURE CustOrderHist @CustomerID nchar(5)
AS
SELECT ProductName, Total=SUM(Quantity)
FROM Products P, [Order Details] OD, Orders O, Customers C
WHERE C.CustomerID = @CustomerID
AND C.CustomerID = O.CustomerID AND O.OrderID = OD.OrderID AND OD.ProductID = P.ProductID
GROUP BY ProductName

Si lo abrimos y le damos click derecho – Execute:

Nos da como resultado:
Running [dbo].[CustOrderHist] ( @CustomerID = ALFKI ).

ProductName                              Total       
---------------------------------------- ----------- 
Aniseed Syrup                            6           
Chai                                     23          
Chartreuse verte                         21          
Escargots de Bourgogne                   40          
Flotemysost                              20          
Grandma's Boysenberry Spread             16          
Lakkalikööri                             15          
Original Frankfurter grüne Soße          2           
Raclette Courdavault                     15          
Rössle Sauerkraut                        17          
Spegesild                                2           
Tofu                                     33          
Vegie-spread                             20          
No rows affected.
(13 row(s) returned)
@RETURN_VALUE = 0
Finished running [dbo].[CustOrderHist].


Del Server Explorer arrastramos el Stored Procedure CustOrderHist hacia el Methods Pane:

Agregamos una página Web DefaultCustOrderList.aspx y colocamos un GridView que llamamos gvCustOrderList.

El código subyacente de la página es:

Partial Class DefaultCustOrderList
    Inherits System.Web.UI.Page

    Protected Sub Page_Load( _
        ByVal sender As Object, _
        ByVal e As System.EventArgs) _
        Handles Me.Load

        Dim db As New dcCustOrderListDataContext
        Dim q = db.CustOrderHist("ALFKI")

        Me.gvCustOrderList.DataSource = q
        Me.gvCustOrderList.DataBind()

    End Sub

End Class

Hoy trabajaremos con un Stored Procedure que tiene un OUTPUT value:
ALTER PROCEDURE dbo.GetProductsByCategory
	(
	@CategoryID int,
	@count int=0 OUTPUT
	)
AS
	select ProductID, ProductName, CategoryID, UnitPrice
	from Products
	where CategoryID = @CategoryID
	order by ProductName
	
	select @count = @@ROWCOUNT
	
	RETURN

Lo creamos y luego lo arrastramos al area de diseño de dcCustOrderList.dbml.
Agregamos un label a la página Web. Yo le puse de ID: lblRowCount

Ajustamos el código para cuando cargue la página:
    Protected Sub Page_Load( _
        ByVal sender As Object, _
        ByVal e As System.EventArgs) _
        Handles Me.Load

        Dim db As New dcCustOrderListDataContext

        Dim rowCount As Integer?

        Dim q = db.GetProductsByCategory(1, rowCount)

        Me.lblRowCount.Text = _
           string.Format( _
           "Registros Encontrados {0}", _
           rowCount)

        Me.gvCustOrderList.DataSource = q
        Me.gvCustOrderList.DataBind()

    End Sub

